HRP 261 SAS LAB THREE, January 28, 2009

Lab Three: importing data, sensitivity and specificity, ROC curves, PROC GPLOT

Lab Objectives

After today’s lab you should be able to:

1. Import a dataset into SAS from excel, using point-and-click.

2. Import a dataset into SAS from excel, using code.

3. Look at data using interactive data analysis.

4. Manipulate data in datasteps.

5. Use PROC FORMAT to label values of a variable in the output.

6. Use PROC SORT to sort data.

7. Use PROC RANK to create categories based on quartiles, tertiles, etc.

8. Calculate sensitivity, specificity, PPV, NPV, and their confidence limits from a 2x2 table using PROC FREQ.

9. Add a title to your results (output).

10. Understand the concept of an ROC curve.

11. Use logistic regression to generate data for an ROC curve.

12. Use PROC GPLOT to plot an ROC curve.

13. If time, run a simulation in SAS.

14. If time, generate a histogram using PROC UNIVARIATE.

 LAB EXERCISE STEPS:

Follow along with the computer in front…

1. Goto the class website: www.stanford.edu/~kcobb/courses/hrp261 --> Download Lab Three DATA onto the desktop (Right click on the link, and then save the data on the desktop as an excel file: psa.xls.)

2. Back in SAS, name a SAS library hrp261 that points to the desktop as follows:

a. Double Click the “New Library” icon (slamming file cabinet) on your menu bar 

b. Type hrp261 in the library Name box.

c. Browse to find your desktop. Then click OK.

d. Highlight the folder extension and use CNTRL C to copy the extension. 

e. Click OK to exit the new library screen.

3. Use your explorer browser to make sure that an hrp261 library was created.

4. Also practice writing the proper SAS code to create a library as follows (use CNTRL V to paste the extension of your desktop).  Notice that you can have more than one library pointing to the same physical folder.

libname repeat 'C:\……paste your extension here...’;

5. Use your explorer browser to make sure that a library named repeat was created.

6. Import the data using point-and-click: 

a. Goto: File--> Import Data-->to open Import Wizard

b. Select Microsoft Excel 97, 2000, or 2002 Workbook (default)--> Next-->

c. Browse to find and select the file psa.xls in your desktop folder. Click Open. Click OK. 

d. Under “what table do you want to import?” leave psa selected-->Next-->

e. Under “Choose the SAS destination” scroll to pick the hrp261 library; then, under member, type: psa to name the dataset hrp261.psa. --->Next-->
f. Browse to find your desktop. Name a file importcode.sas.--->Finish-->
(This last optional step generates the SAS code for importing the psa data, and saves it to a SAS editor file for you—automated programming!)  

7. Use your explorer browser to make sure that a SAS dataset psa was created in the hrp261 library.

8. Return to the SAS enhanced editor window. Then goto “File” on your menu bar, and select “Open Program.” Browse to find the program “importcode.sas” on your desktop. Open it.

Should look like:

PROC IMPORT OUT= HRP261.psa 

            DATAFILE= "C:\….your extenstion…\Desktop\psa.xls" 

            DBMS=EXCEL REPLACE;

     SHEET="psa"; 

     GETNAMES=YES;

     MIXED=NO;

     SCANTEXT=YES;

     USEDATE=YES;

     SCANTIME=YES;

RUN;
Voilà! SAS code is written for you for future reference!  

9. The first thing you should do when you get new data is to look at the data!

Examine the distributions of variables using point-and-click as follows:

a. From the menu select: Solutions(Analysis(Interactive Data Analysis
b. Double click to open: library “hrp261”, dataset “psa”

c. Highlight “capsule” variable from the menu select: Analyze(Distribution(Y)
d. From the menu select: Tables(Frequency Counts
e. Scroll down the open analysis window to examine the frequency counts for capsule (capsule is our outcome variable: capsule=1 if capsule penetration occurred (cancer is more advanced); =0 if not) 

f. Highlight “psa” variable (psa is the diagnostic test) from the menu select: Analyze(Distribution(Y)
g. Place psa and capsule distribution windows side by side. Select the green bar that represents capsule=1; SAS highlights the values of psa of those with capsule=1; allows you to compare the distributions of psa and capsule visually! 
h. Look at the distributions of the other variables, and how they correlate with the distribution of capsule.
i. Use this feature of SAS to familiarize yourself with a new dataset and to check for outliers, missing data, and values that don’t make sense.
10. How good is the PSA test for predicting capsule=1 (the bad outcome)? We could dichotomize the results of the psa test and determine specificity and sensitivity from a two-by-two table as follows: 

*Close your dataset if it’s still open, and then use PROC SORT to sort the data by psa level (note that if your dataset is open, SAS will not sort it!!)

proc sort data=hrp261.psa;

by psa;

run;



/*Break the data into quartiles using PROC RANK*/ 
proc rank data=hrp261.psa out=psa groups=4;

var psa;

ranks psaranks;

run;
/*In a datastep, create a new binary variable “test” that equals 1 if the man is in the top quartile of psa and 0 if he is not.*/

data psa;

set psa;

if psaranks=3 then test=1;

else if psaranks>=0 then test=0;

run;

/*Finally, use PROC FREQ to get sensitivity, specificity, PPV, and NPV. SAS will not automatically generate these values, but since they are just proportions, we can easily generate them. To get the confidence limits for the proportion (for the sensitivity, specificity, etc.), use the “exact binomial” option.*/

/*Sometimes we want to calculate the proportion where test=1 (sensitivity); sometimes we want to calculate the proportion where test=0 (specificity); sometime where capsule=1 (ppv) and sometimes where capsule=0 (npv). SAS will always calculate the proportion with the value “0” (because it comes first numerically), so we need to use a little trickery to get what we want. PROC FORMAT will help...*/

/*PROC FORMAT allows me to assign labels to numerical values. These labels appear in the output (for readability) and can also be used to change the order in tables from 0, 1 to 1, 0, e.g. when we want to calculate the proportion with the value “1” rather than “0”*/

proc format;

value disease

0="no disease"
1="disease"
;

run;
1. SENSITIVITY

Among men who truly have capsule=1, what proportion 

test positive (are in the top quartile of psa)?


proc freq data=psa order=formatted;

      tables test;

      exact binomial;


 where capsule=1;

      title 'Sensitivity';

format test disease.;
run;


  




Binomial Proportion

                                      for test = disease

                              ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                              Proportion (P)               0.4183 sensitivity

                              ASE                          0.0399

                              95% Lower Conf Limit         0.3401  confidence limits

                              95% Upper Conf Limit         0.4965
2. SPECIFICITY

Among men who truly have capsule=0, what proportion 

test negative (are in the lowest 3 quartiles of psa)?
proc freq data=psa;

      tables Test;

      exact binomial;


  where capsule=0;

      title 'Specificity';

run; 

  



    Specificity        

                                       The FREQ Procedure

                                                    Cumulative    Cumulative

                   test    Frequency     Percent     Frequency      Percent

                   ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                      0         196       86.34           196        86.34

                      1          31       13.66           227       100.00


                               Binomial Proportion for test = 0

                             ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                             Proportion (P)                 0.8634

                             ASE                            0.0228

                             95% Lower Conf Limit           0.8188

                             95% Upper Conf Limit           0.9081

                             Exact Conf Limits

                             95% Lower Conf Limit           0.8118

                             95% Upper Conf Limit           0.9053

3. PPV

Among men who test positive, what proportion truly have capsule=1? 

proc freq data=psa order=formatted;

      tables capsule;

      exact binomial;


  where test=1;

      title 'Positive predictive value';


 format capsule disease.;

run;
Positive predictive value                                

                                             capsule

                                                      Cumulative    Cumulative

                  capsule    Frequency     Percent     Frequency      Percent

                  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                   disease         64       67.37            64        67.37

               no disease          31       32.63            95       100.00

                               Binomial Proportion for capsule = 1

                              ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                              Proportion (P)                 0.6737

                              ASE                            0.0481

                              95% Lower Conf Limit           0.5794

                              95% Upper Conf Limit           0.7680
                              Exact Conf Limits

                              95% Lower Conf Limit           0.5698

                              95% Upper Conf Limit           0.7664

I’ll leave NPV for you to do on your own. . . 

11. What if we want to use a continuous predictor or multiple predictors rather than a binary cut-off? Logistic regression…

Though we have not covered logistic regression in class yet, the idea is to predict the probability of a binary outcome as a function of some predictors (here we will use the two diagnostic tests psa and gleason). 

How good of a diagnostic test for identifying capsule=1 (advanced cancer) is my logistic model (gleason + psa)? Make an ROC (Receiver-Operator Characteristic) curve.
The ROC curve plots the false positive rate on the X-axis and (1 - the false negative rate) on the Y-axis as a function of predicted probability (e.g., if I require a predicted probability of 100% to classify a man as having the disease, then my specificity will be perfect, but my sensitivity will be 0).

If the area under the ROC curve is close to 1, you have a very good test. If the area is close to 0.5, you have a lousy test (no better than chance). 

Here’s the code:

proc logistic data = psa;

       model capsule (event = "1") = gleason psa / outRoc = MyRocTable;


 title ' ';
run;

12. Open work.MyRocTable dataset in viewtable mode to get a sense of what SAS is doing

   
Obs     _PROB_    _POS_    _NEG_    _FALPOS_    _FALNEG_    _SENSIT_    _1MSPEC_

          1    0.99639       1      227         0          152       0.00654    0.000000

  ...   293    0.00049     153        0        227          0        1.00000     1.00000


13. Plot sensitivity*(1-specificity) at each cut-off level (293 points here) to make the ROC Curve.
/* make an ROC plot*/

proc gplot data = MyRocTable;

        plot _sensit_  * _1mspec_ ;

run;

quit;

/* use a line instead of dots on the graph*/

proc gplot data = MyRocTable;

        symbol1 i=join v = none c=black line=1;

        plot _sensit_  * _1mspec_ ;

run;

quit;

A good ROC curve should look like (area under the curve better than 50%):

Our ROC curve here is OK, not great:

14. Simulation: simulate the sampling distribution of the odds ratio and of the natural log of the odds ratio using this code:

/** This program simulates the results of 1000 case-control studies, where you can vary the true odds ratio in the population (e.g., the true association between exposure and disease) and the true prevalence of disease in the controls (initially set at 1.0 and 20%)**/ 

/**For each simulated study, SAS draws a sample of 50 cases and 50 controls; determines how many are exposed in each group (using a random binomial function); and calculates an odds ratio (OR) and an lnOR**/

/**The program then plots a histogram of the 1000 simulated ORs (or lnORs)**/ 
%LET OR=1.0;

%LET pe=.20;

%LET NCase=50;

%LET NControl=50;

%LET SIM=1000;

%LET seed=10;


data simulate1;


pe_d=(&pe*&OR)/(&OR*&pe+1-&pe);

    do i=1 to &sim by 1;

            a = ranbin(&seed, &NCase, pe_d);

            b = ranbin(&seed, &NControl, &pe);

            c = &NCase-a;

            d = &NControl-b;

            ORhat = (a*d)/(b*c);




 lnOR = log((a*d)/(b*c));


           output;

    end;

run;

title "Simlulated results from &sim. case-control studies";

proc univariate data=SIMULATE1 noprint;

         pattern1 color = red;


    label orhAT="Observed odds ratio";


    label lnoR= "Observed natural log of the odds ratio";

         var orhat lnor;

         histogram;

run;

NOW try varying the OR, proportion exposed, number of simulations, etc. to see how everything changes!

APPENDIX A:  PROC GPLOT 

Options for plotting symbols in SAS/GRAPH:

Syntax examples:

symbol1 v=star c=yellow h=1 w=1;

symbol2 value=& color=green h=2 w=2;

Options for plotting lines in SAS/GRAPH:

Syntax examples (place within a symbol statement):  

symbol3 v=none c=black w=2 i=join line=5;

Since I want to get the proportion with test=0, I allow SAS to use numerical order, so 0 comes first and 1 second (i.e., I don’t apply a format).





Apply the “disease” format to the variable test.





Order the variables by their labels, rather than numerically. Thus, 1 comes before 0 (d before n).








Name the format anything you want.














Get a histogram from proc univariate.





I’ve fancied it up a bit with a title, labels, and red color.





Example of what happens in 1 loop:





*SAS picks a random variable from a binomial distribution with N=50 and p=.20 This is the number of exposed cases.





*SAS does the same thing to get the number of exposed controls. 





*Unexposed cases = 50 – exposed cases





*Unexposed controls = 50 – exposed controls





*Calculate OR, lnOR


*Output these to a dataset.





Loop through 1000 times and each time output the resulting OR and lnOR


























Tells SAS to use no plotting symbol, and instead connect points (i=join) with a black line. 


line=1 gives a regular solid line


See APPENDIX A for other line options (such as dashes).





Plots sensitivity against 1 minus specificity.


SAS automatically uses a plus-sign for a plotting symbol. See APPENDIX A for symbol options (such as circles).





OutRoc option tells SAS to output the following (into a new dataset):


Start at the highest predicted probability (here: 0.99639). 


Make this the cut-off for classifying a patient as test + (i.e., only patients who’s pred prob>= 0.99639 have a + test).


Using this cutoff, fill in the sensitivity/specificity 2x2 table:


                        Truth�
�



Test �
�
Capsule=1�
Capsule=0�
�
�
+�
1�
0�
�
�
-�
152�
227�
�
Calculate sensitivity= 1/153=.00654


Calculate specificity= 227/227=100%


Interpretation: at such a high cut-off you won’t have any false positives (perfect specificity), but you’ll have tons of false negatives (low sensitivity).


Calculate 1-specificity = 100%-100% = 0.000  (for plotting ROC curve)


Repeat for all unique predicted probabilities in the dataset (here 293). 


As another example, for the lowest predicted probability (.00049):


                        Truth�
�



Test �
�
Capsule=1�
Capsule=0�
�
�
+�
153�
227�
�
�
-�
0�
0�
�
Calculate sensitivity= 153/153=100%


Calculate specificity= 0/227=0%


Interpretation: at such a high cut-off you won’t have any false negatives (perfect sensitivity), but you’ll have tons of false positives (low specificity).


1-specificity= 1-0 = 100%





To call a pre-set variable, add an ampersand to the front.





Set the following parameters at the front of the program, so that you can more easily change them later. 





You must clear the title, or SAS will label your new output with the last table assigned.





Asks to put calculations for an ROC table into a new dataset, which we’re calling “MyRocTable” here.





PPV = 67.4%





Specificity is better….





A sensitivity of 41% is not too great!





out=newdataset : specifies the name of the new dataset that will contain the ranked variable (avoid writing over your old dataset—always precarious!)


groups specifies how many ranks you want (will run from 0 to #groups-1)


ranks statement names a new variable that will contain the ranks—if you omit the ranks statement SAS will write over the old variable with the ranks, and you will lose the original values.


var statement specifies which variable you want ranked. 





**PROC RANK will not work unless you use PROC SORT first to order your data by the variable you want to rank.





Add a title to your output to make your life easier!





restricts the analysis only to the men who have capsule=1 (who truly have the outcome!). So we get the proportion who tested positive among those who actually have the disease.








groups=4 tells SAS you want quartiles.


To get tertiles, use groups=3.


For quintiles, use groups=5.


Etc.

